Agricultural Economics
Series 26

August, 1976

A COMPUTER PROGRAM FOR DETERMINING THE EXPECTED RATE OF RETURN ON THE INVESTMENT IN A COLLEGE EDUCATION



Agricultural Experiment Station
of Auburn University

R.D. Rouse, *Director*

Auburn, Alabama

A COMPUTER PROGRAM FOR DETERMINING THE EXPECTED RATE OF RETURN ON THE INVESTMENT IN A COLLEGE EDUCATION

William E. Hardy, Jr. and John L. Adrian, Jr.*

In recent years, economists, educators, and others concerned with the education process have expressed interest in the private and social returns generated by investment in a college education. Economists have devoted study to these returns and to the related concept of human capital, the productive asset resulting from educating and training the individual [4, p. 571]. Economic analyses have also been directed toward study of the costs of the educational process and determination of how these expenses should be shared by the individual and society [1, 5].

This bulletin presents several basic concepts that should be considered in an education investment analysis. A discussion of the basic returns and costs associated with higher education and a procedure for determining the rate of return on the investment are included. Also, detailed instructions for utilization of a computer program designed to aid in determining the expected rate of return for an individual student's investment in college are presented.

*Assistant Professors in the Department of Agricultural Economics and Rural Sociology, Agricultural Experiment Station, Auburn University.

Rate of Return Analysis

A basic component of all studies related to an investment, whether in human capital or "hard" capital such as machines, buildings, or equipment, is that of the rate of return. When a business manager considers an investment, costs and returns for that particular activity are generally assembled and compared to determine the relative profitability of the investment. A potential student should initiate a similar analysis when considering an investment in higher education.

Educational Returns

In determining the rate of return on educational investment, an individual must estimate all returns expected to be realized as a result of the decision to secure higher education. Returns are grouped into two distinct categories, investment returns and consumption returns. Investment returns refer to monetary rewards or increased earnings that a person might expect through increased skills and/or abilities to produce those goods and services that are valued highly by society. Nonpecuniary or consumption returns, which are perhaps the most important but most difficult to measure, stem from the immediate utility or satisfaction that a person receives during the time of his schooling and the long-run increased satisfaction that accrues during his lifetime because of the educational experience [3, pp. 325- 328].

Educational Costs

Costs of a college education may also be divided into several categories. The major expenses that the student and/or his family

must bear for a college education are tuition, books and supplies, and foregone earnings by the student. The first two components are often referred to as explicit costs since they are obviously costs of attaining higher education. The costs of room and board are not included since a person has to live whether he is in school or not. However, a portion of this expense may be included as a valid cost of education if it is felt that the cost of living in a college community is higher than would be experienced otherwise. This adjustment might be negative if it is felt that the cost of living in the university environment is less expensive.

The foregone earnings component of private educational costs is often overlooked because this amount is not actually paid by check or cash; it is an implicit cost. It is, however, a very important part of the total cost of education, possibly as much as 80% of the annual per student cost in public universities, and should be included in a rate of return analysis since students generally give up full-time employment earnings while in college [2, p. 96].

To determine the rate of return on an individual's investment in higher education, the expected investment returns must be compared to the expected costs. The applicable rate of return is that rate which will make the present value of the expected costs equal to the present value of the projected returns. An alternative formulation would be to determine the discount rate which would make the present value of the net difference between the costs and returns equal to zero. This latter approach may be presented as follows:

$$\sum_{j=1}^{m} \frac{(R_j - C_j)}{(1+i)^j} = 0$$

where

R_j is the return associated with each period, j;

C_j is the cost associated with each period, j;

m is the total number of periods; and

i is the discount rate.

The rate of return estimated by this method would be a conservative underestimate of the true rate since not all returns are included. All costs are contained, but the nonpecuniary or consumption returns that a person receives are not incorporated into the analysis.

The rate of the return on the investment in education is an important subject and should be carefully considered by all college or potential college students. It is, however, recognized by very few. This lack of concern comes as a result of several factors, with the most important being a general lack of knowledge concerning basic economic principles and an overriding desire to attend college for other than monetary reasons.

Because there is a general absence of "personal" rate of return to education analysis on the part of students considering college, a computer program was developed to permit students to ascertain the expected rate of return on their educational investment. This rate is based upon their own expected and projected costs and earnings. The following sections of this report will explain the basic computational procedure, the input required, and the output generated.

Computational Procedure

General Information

The computer program, written in FORTRAN and currently being run on a IBM 370/158, is designed to permit an analysis of rate of return data for several students.[1] Thus, the program is ideal for use within the classroom to supplement a discussion of the costs and returns associated with the investment in education. The output is designed to give each student an individual computation of his rate of return and to produce summary tables giving minimum, maximum, and average rates of return and average expected lifetime earnings with and without college for the total group by sex, academic class, school, and curriculum. These summary tables can be used to allow each student to determine his position relative to his peers and reevaluate his own expectations.

Initial General Input

The first input required is a card describing the data being run, the number of curricula being considered, and the number of schools or colleges being analyzed. This information is given in (12A4,2I2) format and is referred to as "CARD TYPE 1".

The next data needed are the names of each curriculum in (3X,6A4) format with the program currently dimensioned to handle up to ninety (90) curricula, CARD TYPE 2. One card is required for each curriculum identification. Although it is not necessary for these cards to be numbered, consecutive numbering would certainly aid in keeping cards in order and coding of student information.

[1]Copies of the source deck are available upon request from the authors.

Cards giving the identification of the schools or colleges being considered are the next portion of data required, CARD TYPE 3. Space is allocated to handle up to twelve (12) colleges with each identification put on a separate card in (3X,6A4) format. Again, card numbering might aid in keeping cards in order and coding other input.

Individual Student Data

Information for determining the rate of return for each student is obtained from the students. Each student is given a form as illustrated by Schedule 1 and is asked to fill in the appropriate information. The first three lines are self-explanatory with each student giving his name, sex, academic class, school, and curriculum.

Each student provides the expected explicit costs of education in the first column. Up to 13 years of schooling, starting at age 18, are permitted. Entries in this column should be annual estimates of the costs of tuition, books, supplies, and any university community cost of living differential. Scholarships and other forms of educational aid should not be considered in this section.

The data given in the second and third columns reflect the student's income expectations. The second column gives the annual income as expected by the student without education past the high school level, while the third column permits the student to give the earnings, scholarships, and other monetary assistance received while attending school and projected earnings after graduation.

It is necessary for the "years of work" to be the same for both the second and third columns. Also, since the program is designed to

DATA FOR DETERMINATION OF
RETURNS TO COLLEGE EDUCATION

Name _____

Sex _____ Academic Class _____

School _____ Curriculum _____

Age	Cost of Tuition and Books	Expected Income Without College Education	Expected Income With College Education
18			
19			
20			
21			
22			
23			
24			
25			
26			
27			
28			
29			
30			
31-35			
36-40			
41-45			
46-50			
51-55			
56-60			
61-65			
66-70			
71-75			
76-80			

Schedule 1. Example of form used to obtain student information for rate of return analysis.

begin the discount process with the data entries at age 18, better results are obtained if students are asked to assume that they are eighteen years old, just graduated from high school, and are considering the possibility of additional education. With this assumption, students start filling in the information on the first line of the table. If they have already been in school for several years, they should know the information for the first few years of school and have to estimate it for the years past their present position.

Data for each student are submitted with four sets of cards. "CARD TYPE 4", the first of this series, gives the basic information about each student in (40A1,8I5) format. These data are:

Column	Code Name	Information
1-40	Name	Student's name
41-45	NYEARS	Number of years in school (1-13 permitted)
46-50	NRETIR	Retirement age (35+ permitted)
51-55	NCLASS	Code for academic class (1-6 permitted)
56-60	NCURR	Code for curriculum (1-90 permitted)
61-65	NSEX	Code for sex (1=male, 2=female)
66-70	NQT	Code for quarter or semester (1-4 permitted)
71-75	NYR	Year
76-80	NSCH	Code for school or college (1-12 permitted)

The next set of cards, "CARD TYPE 5," gives the student's estimates of his college costs in (8F10.0) format. Up to thirteen values are permitted, with two cards required if more than eight values are given.

"CARD TYPE 6" gives the student's income expectations without college and "CARD TYPE 7" presents the expectations with college. These sets of data are punched in (8F10.0) format with several cards required for each.

Figure 1 depicts the normal deck and order of cards for a rate of return analysis. Card types 4, 5, 6, and 7 are repeated for each student. There is no limit on the number of students that can be considered.

Example Analysis

This section presents a hypothetical example which illustrates actual data input and the output generated by a complete analysis. For the example, it is assumed that there are twenty possible curricula and six possible schools or colleges. The curricula being considered are:

1. Agricultural Economics
2. Agronomy
3. Animal Science
4. Horticulture
5. Building Construction
6. Interior Design
7. Chemistry
8. Geology
9. English
10. Mathematics
11. Accounting
12. Business Administration
13. Economics
14. Management
15. Elementary Education
16. Special Education
17. Aerospace Engineering
18. Civil Engineering
19. Electrical Engineering
20. Mechanical Engineering



Figure 1. Card order for rate of return analysis

The schools or colleges being considered are:

1. Agriculture
2. Architecture
3. Arts and Sciences
4. Business
5. Education
6. Engineering

Data for three hypothetical students are presented on completed Schedules 2, 3, and 4. Schedule 2, giving data for B. Ross, indicates that she plans to attend school for four years, enrolled in English. She plans to work for three years after graduation and then take time to have children and raise them to school age.

T. Jefferson, with data given on Schedule 3, plans to attend school for four years, work for two years and then return to school for two years for a graduate degree. The third column of the table illustrates fairly low income levels during school years and higher levels during non-school years. The low income during the first four years could represent part-time earnings and/or scholarships. The slightly higher figures during graduate school indicate the possibility of an assistantship.

The data for G. Washington in Schedule 4 indicates four years of school in Agronomy with expected retirement from work at age 65.

A listing of the data deck required for the analysis of these three students, the complete output, and a listing of the source program are given in the Appendix. The output section indicates that, based upon the data given by the students, B. Ross could expect an 8% rate of return; T. Jefferson, a 7% rate of return; and G. Washington, a 9% return. The average for these three students is 8% with 7% and 9% being the extreme values.

DATA FOR DETERMINATION OF
RETURNS TO COLLEGE EDUCATION

Name B. Ross

Sex F Academic Class Fr.

School Arts & Sciences Curriculum English

Age	Cost of Tuition and Books	Expected Income Without College Education	Expected Income With College Education
18	750	4000	500
19	775	4000	800
20	800	4000	500
21	800	4100	600
22		4300	8000
23		4500	8300
24		4800	9000
25		0	0
26		0	0
27		0	0
28		0	0
29		0	0
30		0	0
31-35		7300	9000
36-40		7800	9500
41-45		8200	10000
46-50		8500	10500
51-55		8600	11000
56-60			
61-65			
66-70			
71-75			
76-80			

RAISE FAMILY

Schedule 2. Example data for hypothetical student.

DATA FOR DETERMINATION OF
RETURNS TO COLLEGE EDUCATION

Name T. Jefferson

Sex M Academic Class Sr.

School Arch. Curriculum Building Const.

Age	Cost of Tuition and Books	Expected Income Without College Education	Expected Income With College Education
18	800	6000	500
19	800	6000	500
20	825	6200	1500
21	850	6300	1500
22	0	6500	8000
23	0	7000	8400
24	1000	7200	5000
25	1000	7400	5000
26		7500	10500
27		8000	10800
28		8100	11000
29		8200	11300
30		8300	11800
31-35		9000	12000
36-40		9500	12400
41-45		10000	12600
46-50		10500	13000
51-55		10800	13500
56-60		11000	14000
61-65			
66-70			
71-75			
76-80			

Schedule 3. Example data for hypothetical student.

DATA FOR DETERMINATION OF
RETURNS TO COLLEGE EDUCATION

Name G. Washington

Sex M Academic Class Jr.

School Agriculture Curriculum Agronomy

Age	Cost of Tuition and Books	Expected Income Without College Education	Expected Income With College Education
18	800	5000	1000
19	825	5000	1000
20	825	5200	2000
21	900	5200	1500
22		6000	8000
23		6000	8100
24		6500	8300
25		6500	8700
26		6800	9000
27		6800	9000
28		7000	9300
29		7400	9500
30		7600	10000
31-35		8000	10000
36-40		9000	10500
41-45		9500	10800
46-50		9800	10900
51-55		10200	11300
56-60		10500	11800
61-65		11000	12200
66-70			
71-75			
76-80			

dule 4. Example data for hypothetical student.

Summary

The computer program and associated instructions presented in this report have proved to be very useful in presenting the concept of educational investment to students in principles of economics courses. The required input data and result and analysis permit students to comprehend the basic concepts of a rate of return analysis for higher education.

Selected References

[1] Carnegie Commission on Higher Education, 1973. Higher Education: Who Pays? Who Benefits? Who Should Pay? McGraw-Hill Book Co., New Jersey.

[2] Leftwich, R. H. and A. M. Sharp. 1974. Economics of Social Issues. Dallas: Business Publications, Inc.

[3] Peterson, W. L. 1974. Principles of Economics: Micro. Homewood, Illinois: Richard D. Irwin, Inc.

[4] Shultz, T. W. 1960. "Capital Formation by Education." J. of Pol. Econ., vol. 68.

[5] _____. 1967. "The Rate of Return in Allocating Investment Resources to Education." J. Hum. Res.

[6] U. S. Department of Commerce, Bureau of the Census. 1974. Current Population Reports--Consumer Income. U. S. Government Printing Office, Washington, D. C.

APPENDIX

Listing of Input and Output from Example Problem and Source Deck

INPUT

```
 EXAMPLE PROBLEM                                        2006
 1AGRICULTURAL ECONOMICS
 2AGRONOMY
 3ANIMAL SCIENCE
 4HORTICULTURE
 5BUILDING CONSTRUCTION
 6INTERIOR DESIGN
 7CHEMISTRY
 8GEOLOGY
 9ENGLISH
10MATHEMATICS
11ACCOUNTING
12BUSINESS ADMINSTRATION
13ECONOMICS
14MANAGEMENT
15ELEMENTARY EDUCATION
16SPECIAL EDUCATION
17AEROSPACE ENGINEERING
18CIVIL ENGINEERING
19ELECTRICAL ENGINEERING
20MECHANICAL ENGINEERING
 1AGRICULTURE
 2ARCHITECTURE
 3ARTS AND SCIENCES
 4BUSINESS
 5EDUCATION
 6ENGINEERING
 B ROSS                                          4  55   1   9   2   1  76   3
     750       775       800       800
    4000      4000      4000      4100          4300      4500      4800         0
       0         0         0         0             0      7300      7800      8200
    8500      8600
     500       800       500       600          8000      8300      9000         0
       0         0         0         0             0      9000      9500     10000
   10500     11000
 T JEFFERSON                                     8  60   4   5   1  11  76   2
     800       800       825       850             0         0      1000      1000
    6000      6000      6200      6300          6500      7000      7200      7400
    7500      8000      8100      8200          8300      9000      9500     10000
   10500     10800     11000
     500       500      1500      1500          8000      8400      5000      5000
   10500     10800     11000     11300         11800     12000     12400     12600
   13000     13500     14000
 G WASHINGTON                                    4  65   3   2   1   1  76   1
     800       825       825       900
    5000      5000      5200      5200          6000      6000      6500      6500
    6800      6800      7000      7400          7600      8000      9000      9500
    9800     10200     10500     11000
    1000      1000      2000      1500          8000      8100      8300      8700
    9000      9000      9300      9500         10000     10000     10500     10600
   10900     11300     11800     12200
```

OUTPUT

EXPECTED INCOMES WITH AND WITHOUT COLLEGE EDUCATION AND RATE OF RETURN FOR STUDENTS IN EXAMPLE PROBLEM DURING SESSION 1 OF 1976.

THIS REPORT IS FOR B ROSS
WHO EXPECTS TO GO TO SCHOOL FOR 4 YEARS
AND WORK UNTIL THE AGE OF 55. WORKING 34 YEARS AFTER COLLEGE.
THIS STUDENT IS ENROLLED IN ENGLISH

WITH THE EXPECTED INCOMES AND COSTS GIVEN BELOW, B ROSS
WILL HAVE AN APPROXIMATE RATE OF RETURN
ON THE INVESTMENT IN EDUCATION OF 6.0 PERCENT.

AGE	EXPECTED INCOME PER YEAR WITH COLLEGE EDUCATION	EXPECTED INCOME PER YEAR WITHOUT COLLEGE EDUCATION	EXPECTED ADDITIONAL INCOME PER YEAR WITH COLLEGE EDUCATION
18	500.00	4000.00	-4250.00
19	800.00	4000.00	-3975.00
20	500.00	4000.00	-4300.00
21	600.00	4100.00	-4300.00
22	6000.00	4300.00	3700.00
23	8300.00	4500.00	3800.00
24	9000.00	4800.00	4200.00
25	0.0	0.0	0.0
26	0.0	0.0	0.0
27	0.0	0.0	0.0
28	0.0	0.0	0.0
29	0.0	0.0	0.0
30	0.0	0.0	0.0
31-35	9000.00	7300.00	1700.00
36-40	9500.00	7800.00	1700.00
41-45	10000.00	8200.00	1800.00
46-50	10500.00	8500.00	2000.00
51-55	11000.00	8600.00	2400.00
TOTALS	277700.00	231700.00	42875.00

EXPECTED INCOMES WITH AND WITHOUT COLLEGE EDUCATION AND RATE OF RETURN FOR STUDENTS IN EXAMPLE PROBLEM DURING SESSION11 OF 1976.

THIS REPORT IS FOR T JEFFERSON
WHO EXPECTS TO GO TO SCHOOL FOR 3 YEARS
AND WORK UNTIL THE AGE OF 60. WORKING 35 YEARS AFTER COLLEGE.
THIS STUDENT IS ENROLLED IN BUILDING CONSTRUCTION

WITH THE EXPECTED INCOMES AND COSTS GIVEN BELOW, T JEFFERSON
WILL HAVE AN APPROXIMATE RATE OF RETURN
ON THE INVESTMENT IN EDUCATION OF 7.0 PERCENT.

AGE	EXPECTED INCOME PER YEAR WITH COLLEGE EDUCATION	EXPECTED INCOME PER YEAR WITHOUT COLLEGE EDUCATION	EXPECTED ADDITIONAL INCOME PER YEAR WITH COLLEGE EDUCATION
18	500.00	6000.00	-6300.00
19	500.00	6000.00	-6300.00
20	1500.00	6200.00	-5525.00
21	1500.00	6300.00	-5650.00
22	8000.00	6500.00	1500.00
23	8400.00	7000.00	1400.00
24	5000.00	7200.00	-3200.00
25	5000.00	7400.00	-3400.00
26	10500.00	7500.00	3000.00
27	10800.00	8000.00	2600.00
28	11000.00	8100.00	2900.00
29	11300.00	8200.00	3100.00
30	11800.00	8300.00	3500.00
31-35	12000.00	9000.00	3000.00
36-40	12400.00	9500.00	2900.00
41-45	12600.00	10000.00	2600.00
46-50	13000.00	10500.00	2500.00
51-55	13500.00	10800.00	2700.00
56-60	14000.00	11000.00	3000.00
TOTALS	473300.00	396700.00	71325.00

EXPECTED INCOMES WITH AND WITHOUT COLLEGE EDUCATION AND RATE OF RETURN FOR STUDENTS IN EXAMPLE PROBLEM DURING SESSION 1 OF 1976.

THIS REPORT IS FOR G WASHINGTON
WHO EXPECTS TO GO TO SCHOOL FOR 4 YEARS
AND WORK UNTIL THE AGE OF 65, WORKING 44 YEARS AFTER COLLEGE.
THIS STUDENT IS ENROLLED IN AGRONOMY

WITH THE EXPECTED INCOMES AND COSTS GIVEN BELOW, G WASHINGTON
WILL HAVE AN APPROXIMATE RATE OF RETURN
ON THE INVESTMENT IN EDUCATION OF 9.0 PERCENT.

AGE	EXPECTED INCOME PER YEAR WITH COLLEGE EDUCATION	EXPECTED INCOME PER YEAR WITHOUT COLLEGE EDUCATION	EXPECTED ADDITIONAL INCOME PER YEAR WITH COLLEGE EDUCATION
18	1000.00	5000.00	-4800.00
19	1000.00	5000.00	-4825.00
20	2000.00	5200.00	-4025.00
21	1500.00	5200.00	-4600.00
22	8000.00	6000.00	2000.00
23	8100.00	6000.00	2100.00
24	8300.00	6500.00	1800.00
25	8700.00	6500.00	2200.00
26	9000.00	6800.00	2200.00
27	9000.00	6800.00	2200.00
28	9300.00	7000.00	2300.00
29	9500.00	7400.00	2100.00
30	10000.00	7600.00	2400.00
31-35	10000.00	8000.00	2000.00
36-40	10500.00	9000.00	1500.00
41-45	10800.00	9500.00	1300.00
46-50	10900.00	9800.00	1100.00
51-55	11300.00	10200.00	1100.00
56-60	11800.00	10500.00	1300.00
61-65	12200.00	11000.00	1200.00
TOTALS	472900.00	421000.00	48550.00

SUMMARY OF INCOME AND RATE OF RETURN INFORMATION

NUMBER	AVERAGE INCOME WITH COLLEGE EDUCATION	AVERAGE INCOME WITHOUT COLLEGE EDUCATION	AVERAGE RATE OF RETURN	MIN RATE OF RETURN	MAX RATE OF RETURN
3	407966.63	349800.00	8.0	7.0	9.0

SUMMARY OF INCOME AND RATE OF RETURN INFORMATION BY SEX.

CODE	NUMBER	AVERAGE INCOME WITHOUT COLLEGE EDUCATION	AVERAGE INCOME WITHOUT COLLEGE EDUCATION	AVERAGE RATE OF RETURN	MIN RATE OF RETURN	MAX RATE OF RETURN
1	2	473100.00	408850.00	8.0	7.0	9.0
2	1	277700.00	231700.00	8.0	8.0	8.0

SUMMARY OF INCOME AND RATE OF RETURN INFORMATION BY ACADEMIC CLASS.

CODE	NUMBER	AVERAGE INCOME WITH COLLEGE EDUCATION	AVERAGE INCOME WITHOUT COLLEGE EDUCATION	AVERAGE RATE OF RETURN	MIN RATE OF RETURN	MAX RATE OF RETURN
1	1	277700.00	231700.00	8.0	8.0	8.0
3	1	472900.00	421000.00	9.0	9.0	9.0
4	1	473300.00	396700.00	7.0	7.0	7.0

SUMMARY OF INCOME AND RATE OF RETURN INFORMATION BY CURRICULUM.

	NUMBER	AVERAGE INCOME WITH COLLEGE EDUCATION	AVERAGE INCOME WITHOUT COLLEGE EDUCATION	AVERAGE RATE OF RETURN	MIN RATE OF RETURN	MAX RATE OF RETURN
AGRONOMY	1	472900.00	421000.00	9.0	9.0	9.0
BUILDING CONSTRUCTION	1	473300.00	396700.00	7.0	7.0	7.0
ENGLISH	1	277700.00	231700.00	8.0	8.0	8.0

SUMMARY OF INCOME AND RATE OF RETURN INFORMATION BY SCHOOL OR COLLEGE

CODE	NUMBER	AVERAGE INCOME WITH COLLEGE EDUCATION	AVERAGE INCOME WITHOUT COLLEGE EDUCATION	AVERAGE RATE OF RETURN	MIN RATE OF RETURN	MAX RATE OF RETURN
AGRICULTURE	1	472900.00	421000.00	9.0	9.0	9.0
ARCHITECTURE	1	473300.00	396700.00	7.0	7.0	7.0
ARTS AND SCIENCES	1	277700.00	231700.00	8.0	8.0	8.0

SOURCE DECK

```
C      THIS COMPUTER PROGRAM MAY BE USED TO CALCULATE THE
C      RATE OF RETURN TO A COLLEGE EDUCATION FOR INDIVIDUAL
C      STUDENTS.  IT WILL ALSO PRESENT SUMMARY TABLES FOR A
C      SET OF DATA GIVING AVERAGE EXPECTED LIFETIME EARNINGS
C      WITH AND WITHOUT COLLEGE, AND THE AVERAGE,THE MAXIMUM,
C      AND THE MINIMUM RATES OF RETURN FOR THE TOTAL SAMPLE
C      AND BY SEX, ACADEMIC CLASS, CURRICULUM, AND THE SCHOOL
C      OR COLLEGE.
C      BASIC INPUT TO THE PROGRAM IS AS FOLLOWS:
C   CARD TYPE 1
C      NAME OF COURSE---NAMEC( )---IN (12A4) FORMAT.
C      NUMBER OF CURRICULUM CARDS---NCODE---COL. 49 & 50.
C      NUMBER OF SCHOOL CARDS---NSC---COL. 51 & 52.
C   CARD TYPE 2
C      CURRICULUM NAME CARDS---CURRCD( )---IN (3X,6A4) FORMAT.
C           ONE CURRICULUM NAME ON EACH CARD.
C   CARD TYPE 3
C      SCHOOL OR COLLEGE NAME CARDS---SCCOD( )---IN (3X,6A4)
C           FORMAT.  ONE SCHOOL NAME ON EACH CARD.
C   CARD TYPE 4
C      NAME OF STUDENT---NAME( ),
C      EXPECTED NUMBER OF YEARS IN COLLEGE---NYEARS
C           (13 YEARS PERMITTED),
C      EXPECTED RETIREMENT AGE---NRETIRE
C           (AGES OF 35 TO 85 PERMITTED),
C      STUDENT'S ACADEMIC CLASS---NCLASS
C           (6 CODES PERMITTED),
C      STUDENT'S CURRICLUM---NCURR
C            (NCODE CODES PERMITTED),
C      STUDENT'S SEX---NSEX
C            (1=MALE,2=FEMALE),
C      QUARTER OR SEMESTER OF COURSE---NQT
C      YEAR OF COURSE---NYR
C      STUDENT'S SCHOOL---NSCH
C      ---IN (40A1,9I5) FORMAT.
C   CARD TYPE 5
C      ANNUAL COST OF BOOKS, TUITION, AND OTHER COLLEGE COSTS
C           (NOT ROOM AND BOARD) FOR EACH YEAR IN SCHOOL---
C           COLCOST( )---IN (8F10.0) FORMAT.
C           MAXIMUM OF 13 VALUES.
C   CARD TYPE 6
C      ANNUAL EARNINGS EXPECTED BY STUDENT IF HE DOES NOT
C            ATTEND COLLEGE---HISCHOOL( )---IN (8F10.0)
C            FORMAT.
C   CARD TYPE 7
C      ANNUAL EARNINGS EXPECTED BY STUDENT IF HE ATTENDS
C           COLLEGE---COLLEGE( )---IN (8F10.0) FORMAT.
C*************************************************************
C      ONLY ONE OF CARD TYPE 1 IS REQUIRED FOR EACH SET OF
```

```
C            DATA.
C      USE AS MANY OF CARD TYPES 2 & 3 AS NEEDED FOR ALL
C           CURRICULA AND SCHOOLS.
C      ONE OF CARD TYPE 4 IS REQUIRED FOR EACH STUDENT.
C      USE AS MANY OF CARD TYPES 5, 6 AND 7 AS NEEDED FOR ALL
C      COSTS AND EXPECTED INCOMES. BEGIN EACH OF THE THREE
C      TYPES OF DATA ON A NEW CARD.
C      ONE SET OF CARDS TYPES 4,5,6 AND 7 IS REQUIRED FOR
C           EACH STUDENT.
C      IF SUMMARY TABLES BY SEX, ACADEMIC CLASS, CURRICULUM,
C      AND SCHOOL ARE NOT DESIRED, LEAVE THESE FIELDS BLANK
C      FOR EACH STUDENT.
       DIMENSION NAMEC(12),NAME(40),COLCOS(15),HISCH(24),
      1 COLLEG(24),ADINC(24),PV(68),SEX(2,6),CLASS(6,6),CURR(
      290,6),ISEX(2),ICLASS(6),ICURR(90),CURRCD(90,6),SC(12,6)
      3,ISC(12),SCCOD(12,6)
       INTEGER BLANK/' '/
       IRD=5
       READ(5,720) NAMEC,NCODE,NSC
       DO 140 J=1,6
       DO 100 I=1,2
  100  SEX(I,J)=0.0
       DO 110 I=1,6
  110  CLASS(I,J)=0.0
       DO 120 I=1,NCODE
  120  CURR(I,J)=0.0
       DO 130 I=1,12
  130  SC(I,J)=0.0
  140  CONTINUE
       IYC=0
       DO 150 I=1,2
       SEX(I,5)=10.**20
  150  SEX(I,6)=0.0
       DO 160 I=1,6
       CLASS(I,5)=10.**20
  160  CLASS(I,6)=0.0
       DO 170 I=1,NCODE
       CURR(I,5)=10**20
  170  CURR(I,6)=0.0
       DO 180 I=1,NSC
       SC(I,5)=10**20
  180  SC(I,6)=0.0
       TOTNO=0.0
       TOTALH=0.0
       TOTALC=0.0
       TOTALR=0.0
       XMIN=10.**20
       XMAX=0.0
       IRETC=0
```

```
C       READ INPUT DATA.
        READ(5,730)((CURRCD(I,J),J=1,6),I=1,NCODE)
        READ(5,730)  ((SCCOD(I,J),J=1,6),I=1,NSC)
  190 READ(IRD,740,END=500) NAME,NYEARS,NRETIR,NCLASS,NCURR,
     1NSEX,NQT,NYR,NSCH
        IRET=0
  200 IF(NSEX.GT.2) GO TO 210
        IF(NCLASS.GT.6) GO TO 220
        IF(NCURR.GT.NCODE) GO TO 230
        IF(NSCH.GT.NSC) GO TO 240
        GO TO 260
  210 WRITE(6,750) NAME
        READ(IRD,740,END=500) NAME,NYEARS,NRETIR,NCLASS,NCURR,
     1NSEX,NQT,NYR,NSCH
        IF(NAME(1).EQ.BLANK) GO TO 250
        GO TO 200
  220 WRITE(6,760) NAME
        READ(IRD,740,END=500) NAME,NYEARS,NRETIR,NCLASS,NCURR,
     1NSEX,NQT,NYR,NSCH
        IF(NAME(1).EQ.BLANK) GO TO 250
        GO TO 200
  230 WRITE(6,770) NAME
        READ(IRD,740,END=500) NAME,NYEARS,NRETIR,NCLASS,NCURR,
     1NSEX,NQT,NYR,NSCH
        IF(NAME(1).EQ.BLANK) GO TO 250
        GO TO 200
  240 WRITE(6,780) NAME
        READ(IRD,740,END=500) NAME,NYEARS,NRETIR,NCLASS,NCURR,
     1NSEX,NQT,NYR,NSCH
        IF(NAME(1).EQ.BLANK) GO TO 250
        GO TO 200
  250 READ(IRD,740,END=500) NAME,NYEARS,NRETIR,NCLASS,NCURR,
     1NSEX,NQT,NYR,NSCH
        IF(NAME(1).EQ.BLANK) GO TO 250
        GO TO 200
  260 PVAD=0.0
        TOTHI=0.0
        TOTCO=0.0
        TOTAD=0.0
C       NWORK IS THE NUMBER OF WORKING YEARS.
C       NINC IS THE NUMBER OF LINES FILLED IN ON HANDOUT.
C             (THE NUMBER OF DATA ELEMENTS)
C       NTOT IS THE NUMBER OF YEARS FROM AGE 18 UNTIL RETIRE.
        NTOT=NRETIR-17
        NWORK=NTOT-NYEARS
        NINC=(NRETIR/5)+7
        READ(IRD,790)(COLCOS(I),I=1,NYEARS)
        READ(IRD,790)(HISCH(I),I=1,NINC)
        READ(IRD,790)(COLLEG(I),I=1,NINC)
```

```
      DO 270 I=1,NYEARS
      ADINC(I)=COLLEG(I)-(COLCOS(I)+HISCH(I))
  270 CONTINUE
      NY=NYEARS+1
      DO 280 I=NY,NINC
      ADINC(I)=COLLEG(I)-HISCH(I)
  280 CONTINUE
      DO 300 I=1,NINC
      IF(I.GT.13) GO TO 290
      TOTHI=TOTHI+HISCH(I)
      TOTCO=TOTCO+COLLEG(I)
      TOTAD=TOTAD+ADINC(I)
      GO TO 300
  290 TOTHI=TOTHI+HISCH(I)*5.0
      TOTCO=TOTCO+COLLEG(I)*5.0
      TOTAD=TOTAD+ADINC(I)*5.0
  300 CONTINUE
      DO 310 I=1,13
      PV(I)=ADINC(I)
  310 CONTINUE
      J=-1
      DO 330 I=14,NINC
      K=I+J
      DO 320 L=1,5
      KJ=K+L
      PV(KJ)=ADINC(I)
  320 CONTINUE
      J=J+4
  330 CONTINUE
      DO 340 I=1,NTOT
      IF(PV(I).LT.0.0) GO TO 350
  340 CONTINUE
      IRETC=IRETC+1
      IRET=1
      GO TO 400
C     CALCULATION OF RATE OF RETURN ON INVESTMENT IN
C           EDUCATION.
C     XINT IS THE DISCOUNT RATE.
  350 XINT=0.0
      PVAD1=99999999
      DO 370 I=1,200
      PVAD=0.0
      DO 360 J=1,NTOT
      PVAD=PVAD+PV(J)/(1.0+XINT)**J
  360 CONTINUE
      APVAD=ABS(PVAD)
      IF(PVAD.LE.0..AND.APVAD.LE.PVAD1) GO TO 390
      IF(PVAD.LE.0..AND.APVAD.GE.PVAD1) GO TO 380
      PVAD1=PVAD
```

```
      XINT=XINT+.01
  370 CONTINUE
  380 XINT=XINT-.01
  390 CONTINUE
      XINT=XINT*100.0
  400 WRITE(6,800) NAMEC,NQT,NYR
      WRITE(6,810) NAME,NYEARS,NRETIR,NWORK,(CURRCD(NCURR,J)
     1,J=1,6)
      IF(IRET.EQ.1) GO TO 410
      WRITE(6,820) NAME,XINT
      GO TO 420
  410 WRITE(6,830) NAME
  420 WRITE(6,840)
      DO 430 I=1,13
      IAGE=I+17
  430 WRITE(6,850) IAGE,COLLEG(I),HISCH(I),ADINC(I)
      J=0
      DO 440 I=14,NINC
      IAGE=I+17+J
      IAGE1=IAGE+4
      J=J+4
  440 WRITE(6,860) IAGE,IAGE1,COLLEG(I),HISCH(I),ADINC(I)
      WRITE(6,870) TOTCO,TOTHI,TOTAD
  450 CONTINUE
C      BUILD SUMMARY TABLES.
      IF(IRET.EQ.1) GO TO 190
      TOTNO=TOTNO+1.0
      TOTALC=TOTALC+TOTCO
      TOTALH=TOTALH+TOTHI
      TOTALR=TOTALR+XINT
      IF(XINT.LE.XMIN) XMIN=XINT
      IF(XINT.GE.XMAX) XMAX=XINT
      IF(NSEX.EQ.0) GO TO 460
      SEX(NSEX,1)=SEX(NSEX,1)+1.0
      SEX(NSEX,2)=SEX(NSEX,2)+TOTCO
      SEX(NSEX,3)=SEX(NSEX,3)+TOTHI
      SEX(NSEX,4)=SEX(NSEX,4)+XINT
      IF(XINT.LE.SEX(NSEX,5)) SEX(NSEX,5)=XINT
      IF(XINT.GE.SEX(NSEX,6)) SEX(NSEX,6)=XINT
  460 IF(NCLASS.EQ.0) GO TO 470
      CLASS(NCLASS,1)=CLASS(NCLASS,1)+1.
      CLASS(NCLASS,2)=CLASS(NCLASS,2)+TOTCO
      CLASS(NCLASS,3)=CLASS(NCLASS,3)+TOTHI
      CLASS(NCLASS,4)=CLASS(NCLASS,4)+XINT
      IF(XINT.LE.CLASS(NCLASS,5)) CLASS(NCLASS,5)=XINT
      IF(XINT.GE.CLASS(NCLASS,6)) CLASS(NCLASS,6)=XINT
  470 IF(NCURR.EQ.0) GO TO 480
      CURR(NCURR,1)=CURR(NCURR,1)+1
      CURR(NCURR,2)=CURR(NCURR,2)+TOTCO
```

```
      CURR(NCURR,3)=CURR(NCURR,3)+TOTHI
      CURR(NCURR,4)=CURR(NCURR,4)+XINT
      IF(XINT.LE.CURR(NCURR,5)) CURR(NCURR,5)=XINT
      IF(XINT.GE.CURR(NCURR,6)) CURR(NCURR,6)=XINT
  480 IF(NSCH.EQ.0) GO TO 490
      SC(NSCH,1)=SC(NSCH,1)+1
      SC(NSCH,2)=SC(NSCH,2)+TOTCO
      SC(NSCH,3)=SC(NSCH,3)+TOTHI
      SC(NSCH,4)=SC(NSCH,4)+XINT
      IF(XINT.LE.SC(NSCH,5)) SC(NSCH,5)=XINT
      IF(XINT.GE.SC(NSCH,6)) SC(NSCH,6)=XINT
  490 GO TO 190
C     WRITE SUMMARY TABLES.
  500 WRITE(6,880)
      ITOTAL=TOTNO+.5
      TOTALH=TOTALH/TOTNO
      TOTALC=TOTALC/TOTNO
      TOTALR=TOTALR/TOTNO
      WRITE(6,890)
      WRITE(6,900) ITOTAL,TOTALC,TOTALH,TOTALR,XMIN,XMAX
      IF(SEX(1,1).EQ.0.0.AND.SEX(2,1).EQ.0.0) GO TO 550
      WRITE(6,910)
      WRITE(6,920)
      DO 520 I=1,2
      IF(SEX(I,1).EQ.0.0) SEX(I,5)=0.0
      IF(SEX(I,1).EQ.0.0) GO TO 520
      DO 510 J=2,4
  510 SEX(I,J)=SEX(I,J)/SEX(I,1)
  520 CONTINUE
      DO 530 I=1,2
  530 ISEX(I)=SEX(I,1)+.5
      DO 540 I=1,2
      IF(ISEX(I).EQ.0) GO TO 540
      WRITE(6,930) I,ISEX(I),(SEX(I,J),J=2,6)
  540 CONTINUE
  550 MAX=0
      DO 560 I=1,6
  560 IF(CLASS(I,1).GT.0.0) MAX=I
      IF(MAX.EQ.0) GO TO 610
      DO 580 I=1,MAX
      IF(CLASS(I,1).EQ.0.0) GO TO 580
      DO 570 J=2,4
  570 CLASS(I,J)=CLASS(I,J)/CLASS(I,1)
  580 CONTINUE
      DO 590 I=1,MAX
  590 ICLASS(I)=CLASS(I,1)+.5
      WRITE(6,940)
      WRITE(6,920)
      DO 600 I=1,MAX
```

```
      IF(CLASS(I,1).EQ.0.0) GO TO 600
      WRITE(6,930) I,ICLASS(I),(CLASS(I,J),J=2,6)
  600 CONTINUE
  610 MAX=0
      DO 620 I=1,NCODE
      IF(CURR(I,1).GT.0.0) MAX=I
  620 CONTINUE
      IF(MAX.EQ.0) GO TO 670
      DO 640 I=1,MAX
      IF(CURR(I,1).EQ.0.0) GO TO 640
      DO 630 J=2,4
  630 CURR(I,J)=CURR(I,J)/CURR(I,1)
  640 CONTINUE
      DO 650 I=1,MAX
  650 ICURR(I)=CURR(I,1)+.5
      WRITE(6,950)
      WRITE(6,960)
      DO 660 I=1,MAX
      IF(CURR(I,1).EQ.0.0) GO TO 660
      WRITE(6,970) (CURRCD(I,J),J=1,6),ICURR(I),(CURR(I,J),
     1J=2,6)
  660 CONTINUE
  670 WRITE(6,980)
      WRITE(6,920)
      DO 680 I=1,NSC
  680 ISC(I)=SC(I,1)+.5
      DO 700 I=1,NSC
      IF(SC(I,1).EQ.0) GO TO 700
      DO 690 J=2,4
  690 SC(I,J)=SC(I,J)/SC(I,1)
  700 CONTINUE
      DO 710 I=1,NSC
      IF(SC(I,1).EQ.0.0) GO TO 710
      WRITE(6,970) (SCCOD(I,J),J=1,6),ISC(I),(SC(I,J),J=2,6)
  710 CONTINUE
      WRITE(6,990) IRETC
  720 FORMAT(12A4,2I2)
  730 FORMAT(3X,6A4)
  740 FORMAT(40A1,8I5)
  750 FORMAT(1H1,5X,'INVALID VALUE FOR SEX FOR ',40A1)
  760 FORMAT(1H1,5X,'INVALID VALUE FOR ACADEMIC CLASS FOR ',
     140A1)
  770 FORMAT(1H1,5X,'INVALID VALUE FOR CURRICULUM FOR',40A1)
  780 FORMAT(1H1,5X,'INVALID VALUE FOR SCHOOL FOR ',40A1)
  790 FORMAT(8F10.0)
  800 FORMAT(1H1///5X' EXPECTED INCOMES WITH AND WITHOUT ',
     1'COLLEGE EDUCATION AND RATE'/5X,'OF RETURN FOR STUDEN'
     2,'TS IN ',12A4/5X,'DURING SESSION',I2,' OF 19',I2,'.')
  810 FORMAT(///5X,'THIS REPORT IS FOR ',40A1/5X,'WHO EXPEC'
```

```
     1,'TS TO GO TO SCHOOL FOR ',I2,' YEARS' /5X,'AND WORK U'
     2,'NTIL THE AGE OF ',I2,', WORKING ',I2,' YEARS AFTER'
     3,' COLLEGE.'/5X,'THIS STUDENT IS ENROLLED IN ',6A4)
 820 FORMAT(///5X,'WITH THE EXPECTED INCOMES AND COSTS ',
     1'GIVEN BELOW, ',40A1/5X,'WILL HAVE AN APPROXIMATE RATE'
     2,' OF RETURN'/5X,'ON THE INVESTMENT IN EDUCATION OF ',
     3F5.1,' PERCENT.')
 830 FORMAT(///5X,'WITH THE EXPECTED INCOMES AND COSTS ',
     1'GIVEN BELOW, ',40A1/5X,'WILL HAVE AN INFINITE RATE'
     2,' OF RETURN'/5X,'ON THE INVESTMENT IN EDUCATION.')
 840 FORMAT(///T62,'EXPECTED'/T6,'AGE',T20,'EXPECTED',T41
     1,'EXPECTED',T61,'ADDITIONAL'/T16,'INCOME PER YEAR',T37
     2,'INCOME PER YEAR',T58,'INCOME PER YEAR'/T18,'WITH CO'
     3,'LLEGE ',   T37,'WITHOUT COLLEGE',T60,'WITH COLLEGE'/T
     420,'EDUCATION',T41,'EDUCATION',T62,'EDUCATION')
 850 FORMAT(T7,I2,T19,F9.2,T40,F9.2,T62,F9.2)
 860 FORMAT(T5,I2,'-',I2,T19,F9.2,T40,F9.2,T62,F9.2)
 870 FORMAT(/T5,'TOTALS',T18,F10.2,T39,F10.2,T61,F10.2)
 880 FORMAT(1H1///T30,'SUMMARY OF INCOME AND RATE OF RETUR'
     1,'N INFORMATION')
 890 FORMAT(///T28,'AVERAGE INCOME',T48,'AVERAGE INCOME'/T2
     19,'WITH COLLEGE',T48,'WITHOUT COLLEGE',T72,'AVERAGE',
     2T91,'MIN',T108,'MAX'/T21,'NUMBER',T31,'EDUCATION',T50,
     3'EDUCATION',T68,'RATE OF RETURN',T86,'RATE OF RETURN',
     4T103,'RATE OF RETURN')
 900 FORMAT(/T22,I3,T30,F10.2,T49,F10.2,T73,F4.1,T90,F4.1,
     1T106,F5.1)
 910 FORMAT(1H1//T30,'SUMMARY OF INCOME AND RATE OF RETURN',
     1' INFORMATION BY SEX.')
 920 FORMAT(///T28,'AVERAGE INCOME',T48,'AVERAGE INCOME'/
     1T29,'WITH COLLEGE',T48,'WITHOUT COLLEGE',T72,'AVERAGE'
     2,T91,'MIN',T108,'MAX'/T14,'CODE',T21,'NUMBER',T31,'EDU'
     3,'CATION',T50, 'EDUCATION',T68,'RATE OF RETURN',T86,'R'
     4,'ATE OF RETURN',T103,'RATE OF RETURN')
 930 FORMAT(/T15,I2,T22,I3,T30,F10.2,T49,F10.2,T73,F4.1,T90
     1,F4.1,T106,F5.1)
 940 FORMAT(1H1//T30,'SUMMARY OF INCOME AND RATE OF RETURN'
     1,' INFORMATION BY ACADEMIC CLASS.')
 950 FORMAT(1H1//T30,'SUMMARY OF INCOME AND RATE OF RETURN'
     1,' INFORMATION BY CURRICULUM.')
 960 FORMAT(///T28,'AVERAGE INCOME',T48,'AVERAGE INCOME'/
     1T29,'WITH COLLEGE',T48,'WITHOUT COLLEGE',T72,'AVERAGE'
     2,T91,'MIN',T108,'MAX'/T21,'NUMBER' ,T31,'EDUCATION',
     3T68 ,'RATE OF RETURN' ,T86,'RATE OF RETURN',T103,'RAT'
     4,'E OF RETURN')
 970 FORMAT(/T2,6A4,T23,I3,T30,F10.2,T49,F10.2,T72,F5.1,T89
     1,F5.1,T106,F5.1)
 980 FORMAT(1H1///T30,'SUMMARY OF INCOME AND RATE OF RETURN'
     1,' INFORMATION BY SCHOOL OR COLLEGE')
```

```
 990 FORMAT(///'A TOTAL OF',I3,' STUDENTS HAVE AN INFINITE'
    1,' RATE OF RETURN ON THEIR INVESTMENT IN EDUCATION.')
1000 STOP
     END
```